Κεφάλαιο 6ο: Όρια ακολουθιών και συναρτήσεων. Ακρότατα συναρτήσεων. Σειρές Taylor.

Η *εντολή με την οποία βρίσκουμε το* όριο *μιας ακολουθίας* είναι *η* Limit.

```
? Limit
```

```
Limit[expr, x->x0] finds the
   limiting value of expr when x approaches x0. More...
```

Με ??Limit *βρίσκουμε τα χαρακτηριστικά της εντολής* Limit

```
?? Limit
```

```
Limit[expr, x->x0] finds the
   limiting value of expr when x approaches x0. More...
```

```
Attributes[Limit] = {Listable, Protected}

Options[Limit] = {Analytic → False, Direction → Automatic}
```

Το protected *σημαίνει* ότι *δεν μπορούμε να την αλλάξουμε* ενώ *το* Listable *σημαίνει* ότι *μπορεί να εφαρμοστεί συγχρόνως σε πολλές ακολουθίες δηλ. σε μια ολόκληρη λίστα απο ακολουθίες*. Π.*χ*

$$\text{Limit}\left[\left\{\frac{\text{Log}[n]}{n}, \sqrt[n]{n}, \frac{1-n^3}{n^2}\right\}, n \to \infty\right]$$

```
{0, 1, -∞}
```

Όπως *βλέπουμε η* $\frac{1-n^3}{n^2}$ *είναι αποκλίνουσα*. Εάν *η ακολουθία* δεν έχει όριο αλλά έχει δύο ή *περισσότερες συγκλίνουσες υποακολουθίες τότε απαντάει με ένα μήνυμα της μορφής* Interval[{a,b}] *που σημαίνει* ότι *υπάρχουν κάποιες πιθανές λύσεις*(όρια) *στο διάστημα* [*α*,*β*]. *π.χ*

$$\text{Limit}\left[\frac{1}{n} + \text{Sin}\left[\frac{n\,\pi}{2}\right] + \text{Cos}\left[\frac{n\,\pi}{2}\right], n \to \infty\right]$$

```
Interval[{-2, 2}]
```

Η *παραπάνω ακολουθία* έχει *δυο οριακούς αριθμούς το* 1 *και το* -1. Άς *κάνουμε και* ένα *διάγραμμα να το δούμε*. Θα χρησιμοποιήσουμε την LabelListPlot αντί της συνηθισμένης Plot διότι στην περίπτωσή μας θέλουμε απλώς να σχεδιάσουμε κάποιος όρους(σημεία) και όχι γραφική παράσταση κάποιας συνάρτησης. Η LabelListPlot σχεδιάζει και συγχρόνως απαριθμεί την ακολουθία. Θα πρέπει όμως στην αρχή να καλέσουμε το ειδικό πακέτο εντολών Graphics`Graphics` που την περιέχει. Το κάλεσμα του πακέτου γίνεται με το <<(διπλό <). Θα πρέπει να έχουμε υπόψη μας ότι κάποιες ειδικές εντολές δεν είναι ετοιμοπαράδοτες από το *Mathematica*. Αυτό γίνεται διότι δεν είναι σωστό να βαραίνει η μνήμη του υπολογιστεί με πάρα πολλές εντολές. Όσες λοιπόν εντολές δεν είναι ετοιμοπαράδοτες θα πρέπει να βρούμε σε πιο πακέτο ανήκουν έτσι ώστε πρώτα να ανοίξουμε το πακέτο και μετά να τις χρησιμοποιήσουμε. Το Help είναι πολύ καλό ώστε να μας οδηγήσει να βρούμε για κάθε ειδική εντολή το δικό της πάκετο.

```
Remove[LabeledListPlot]

oiPrwtoi20oroi = Table[1/n + Sin[n π/2] + Cos[n π/2], {n, 20}];
<< Graphics`Graphics`
LabeledListPlot[oiPrwtoi20oroi]
```



- Graphics -

Επειδή δεν απεικονίζονται όλα *τα σημεία σωστά, αλλάζουμε* λίγο *τα χαρακτηριστικά της* LabeledListPlot:

```
Remove[LabeledListPlot]

oiPrwtoi20oroi = Table[1/n + Sin[n π/2] + Cos[n π/2], {n, 20}];
<< Graphics`Graphics`
LabeledListPlot[oiPrwtoi20oroi,
 PlotRange → All, AxesOrigin → {0, 0}, AspectRatio → 0.8]
```



- Graphics -

Για τις επιλογές PlotRange, PlotRange και AspectRatio θα μάθουμε στο αντίστοιχο κεφάλαιο με τα γραφήματα. Μπορείτε να πληροφορηθείτε περισσότερα και από το Help.

Άσκηση. Μάθετε ποιές άλλες ειδικές συναρτήσεις περιέχει το σπουδαίο πακέτο για γραφήματα Graphics`-Graphics` .

Υπάρχει και άλλος ένας τρόπος να δούμε τα παραπάνω σημεία με την χρήση της TextListPlot

$$\text{TextListPlot}\left[\text{Table}\left[\frac{1}{n}+\text{Sin}\left[\frac{n\,\pi}{2}\right]+\text{Cos}\left[\frac{n\,\pi}{2}\right],\ \{n,\ 20\}\right]\right]$$



- Graphics -

Η TextListPlot είναι όπως η LabeledListPlot χωρίς όμως να μπαίνουν οι κουκίδες στα σημεία. Επειδή ανήκει στο ίδιο πακέτο με αυτήν δεν χρειάζεται να ξανακαλέσουμε για δεύτερη φορά το πακέτο. ***Υπάρχει και μια χειρότερη περίπτωση που ενώ υπάρχει το όριο το*** Mathematica 4(οι επόμενες εκδόσεις του *Mathematica* είναι σαφώς βελτιωμένες)***αδυνατεί να απαντήσει*** ή ***να δώσει λάθος απάντηση π.χ***

$$\text{Limit}\left[\left\{\frac{n!}{n^n},\ \sqrt[n]{n!}\right\},\ n\to\infty\right]$$

Series::esss : Essential singularity encountered in Gamma$[\frac{1}{n}+1+O[n]^3]$.

Series::esss : Essential singularity encountered in Gamma$[\frac{1}{n}+1+O[n]^3]$.

Series::esss : Essential singularity encountered in Gamma$[\frac{1}{n}+1+O[n]^3]$.

General::stop : Further output of Series::esss will be suppressed during this calculation.

$$\{\text{Limit}[n^{-n}\,n!,\ n\to\infty],\ \text{Limit}[n!^{\frac{1}{n}},\ n\to\infty]\}$$

Είναι ***γνωστό*** ότι ***η πρώτη*** είναι ***μηδενική και η δεύτερη αποκλίνουσα στο*** $+\infty$.

```
ListPlot[Table[{n, n^-n n!}, {n, 100}],
  PlotLabel -> "οι όροι της ακολουθίας n!/n^n"]
```



- Graphics -

Να αναφέρουμε εδώ ότι η ListPlot σχεδιάζει μόνο τους όρους της ακολουθίας και είναι ετοιμοπαράδοτη δηλ. δεν χρειάζεται κανένα ειδικό πακέτο για να την καλέσουμε. ***Ας σχεδιάσουμε τώρα την*** $n!^{\frac{1}{n}}$. Η $n!^{\frac{1}{n}}$ ***συμπεριφέρεται γραμμικά***(!) ***με κλίση "περίπου" είναι ίση με*** 1/e. ***Για να το αποδείξουμε αυτό***(***γραφική απόδειξη***!) ***σχεδιάζουμε και την*** ακολουθία $\frac{n}{e}$ ***με διαφορετικά χρώματα για να δούμε την διαφορά***. Σχεδιάζουμε μόνο τους 70 πρώτους όρους απο κάθε ακολουθία.

```
timesToyn = 70;
Remove[DisplayTogether]
<< Graphics`Graphics`
p3 =
 DisplayTogether[ListPlot[Table[{n, n!^(1/n)}, {n, 1, timesToyn}],
   PlotStyle → RGBColor[1, 0, 0]],
  ListPlot[Table[{n, n/E}, {n, 1, timesToyn}],
   PlotStyle → RGBColor[0, 0, 1]]]
```



- Graphics -

Γενικά η χρήση των γραφικών παραστάσεων είναι ένα χρήσιμο οπτικό *εργαλείο*. Η DisplayTogether χρειάζεται το πακέτο Graphics`Graphics`. Βέβαια το έχουμε καλέσει προηγουμένως οπότε δεν ήταν απαραίτητο να το ξανακαλέσουμε εδώ. (Αν όμωςχρησιμοποιήσουμε την DisplayTogether χωρίς να έχουμε πρώτα καλέσει το πακέτο της Graphics`Graphics`, τότε δεν θα βγάλει τίποτα! Σε τέτοιες περιπτώσεις θα πρέπει να κάνουμε την επόμενη διαδικασία : πρώτα θα γράψουμε Remove[DisplayTogether] και στην συνέχεια θα καλέσουμε το πακέτο της και τότε πια μπορούμε να την χρησιμοποιήσουμε χωρίς πρόβλημα) Για την DisplayTogether μπορείτε να μάθετε περισσότερα απο το Help. Βασικά συνδυάζει γραφικές παραστάσεις. Εναλλακτικά θα μπορούσατε να χρησιμοποιήσετε την Show. *Όπου είναι εφικτό θα δίνουμε και μια κατάλληλη γραφική παράσταση*. Για τα όρια συναρτήσεων θα μιλήσουμε σε επόμενη ενότητα.

6.1 Αθροίσματα, Γινόμενα,Σειρές

Τό *άθροισμα δίνεται απο την συνάρτηση* Sum. *Τα* όρια *του αθροίσματος δίνονται με κάποιες τιμές στα* min *και* max. Το d , *αν υπάρχει παριστάνει το βήμα που αυξάνονται οι τιμές της μεταβλητής* i. *Συνοψίζοντας έχουμε τις παρακάτω μορφές*

Sum[f,{i,max}] ή Sum[f,{i,min,max}] ή Sum[f,{i,min,max,d}]

Παραδείγματα:

```
seira = Sum[x^i/i, {i, 1, 10, 2}]
```

$$x+\frac{x^3}{3}+\frac{x^5}{5}+\frac{x^7}{7}+\frac{x^9}{9}$$

```
seira^2
```

$$\left(x+\frac{x^3}{3}+\frac{x^5}{5}+\frac{x^7}{7}+\frac{x^9}{9}\right)^2$$

```
Expand[seira^2]
```

$$x^2+\frac{2x^4}{3}+\frac{23x^6}{45}+\frac{44x^8}{105}+\frac{563x^{10}}{1575}+\frac{124x^{12}}{945}+\frac{143x^{14}}{2205}+\frac{2x^{16}}{63}+\frac{x^{18}}{81}$$

Όπως *βλέπουμε μια σειρά μπορούμε την χειριστούμε* όπως *θέλουμε*. Π.*χ να βρούμε το τετράγωνό της και επίσης μπορούμε και να την παραγωγήσουμε* ή *να ολοκληρώσουμε*!

```
D[seira, {x, 2}](* deyterh paragwgos ws pros x *)
integ = Integrate[seira, x](* aoristo oloklhrwma ws pros x*)
```

$$2x+4x^3+6x^5+8x^7$$

$$\frac{x^2}{2}+\frac{x^4}{12}+\frac{x^6}{30}+\frac{x^8}{56}+\frac{x^{10}}{90}$$

Η γνωστή μας απο παλαιότερα *συνάρτηση* Coefficient *είναι πολύ χρήσιμη διότι όταν το ανάπτυγμα της σειράς είναι αρκετά μεγάλο μπορεί να μας βρεί τον συντελεστή μιας δύναμης* x^n *μιας μεταβλητής* x . Π.*χ*

```
Coefficient[integ, x, 10]
```

$$\frac{1}{90}$$

Άλλες *παραλαγές του* Coefficient *μπορείτε να βρείτε πατώντας* F1(για να καλέσουμε το Help) *κατα τα γνωστά*.

Πρέπει σε αυτό το σημείο να τονίσουμε ότι η Plus είναι πολύ κατώτερη απο την Sum διότι δεν ξέρει να αθροίζει αθροιστέους που περιέχουν παραμέτρους π.χ

```
Plus @@ Table[i^2, {i, 1, 5}]
Plus @@ Table[i^2, {i, 1, n}]

55
```

```
Table::iterb : Iterator {i, 1, n} does not have appropriate bounds.
```

$\{i + i^2, 1 + i^2, i^2 + n\}$

Βλέπουμε ότι το πρώτο άθροισμα το βρήκε το δεύτερο όχι. Να θυμίσουμε ότι το @@ χρησιμοποιείται αντί της Apply.

Παρακάτω παραθέτονται τα μερικά αθροίσματα δυνάμεων του i *σε μορφή πίνακα και η γνωστή μας γεωμετρική πρόοδο με λόγο ω.*

```
Table[{"για m=", m, " άθροισμα σειράς ίσο με", Sum[i^m, {i, 1, n}]},
  {m, 1, 3}] // TableForm
```

για m=	1	άθροισμα σειράς ίσο με	$\frac{1}{2} n (1 + n)$
για m=	2	άθροισμα σειράς ίσο με	$\frac{1}{6} n (1 + n) (1 + 2 n)$
για m=	3	άθροισμα σειράς ίσο με	$\frac{1}{4} n^2 (1 + n)^2$

```
Sum[a ω^(i-1), {i, 1, n}]
```

$$\frac{a (-1 + \omega^n)}{-1 + \omega}$$

Για να βρούμε αν η παραπάνω γεωμετρική σειρά συγκλίνει θέτουμε n=∞

```
Sum[a ω^(i-1), {i, 1, ∞}]
```

$$-\frac{a}{-1 + \omega}$$

Πρέπει να προσέξουμε ότι σιωπηλά η Sum *υποθέτει ότι ο παρανομαστής* $-1+\omega$ *έχει την ιδιότητα* $|\omega|<1$ *διότι αλλιώς δεν συγλίνει!* Ας *δούμε τι γίνεται σε αντίθετη περίπτωση* π.χ. για ω=1.5 και a=1

```
Sum[(1.5)^(i-1), {i, 1, ∞}]
```

```
Sum::div : Sum does not converge.
```

$$\sum_{i=1}^{\infty} 1.5^{i-1}$$

Ας δούμε λίγο και την γραφική παράσταση των μερικών αθροισμάτων για να επιβεβαιώσουμε οπτικά την απόκλιση

```
ListPlot[Table[{n, Sum[(1.5)^(i-1), {i, 1, n}]}, {n, 10}],
 PlotStyle → AbsolutePointSize[4]]
```




```
- Graphics -
```

Σχόλιο: Το *σύμβολο του απείρου μπορεί να εισαχθεί γράφοντας* \[Infinity] (*μεταξύ του* Infinity *και*] *δεν πρέπει να υπάρχει κενό γιατί δεν θα μεταραπεί αυτόματα σε* ∞) ή *πατώντας το πλήκτρο* Esc *και μετά γράφοντας* inf *και ξαναπατώντας το* Esc *χρησιμοποιώντας την βασική παλέτα* 3 BasicInput. Όμοια *το* $\sum_{n=0}^{m} \frac{1}{f[n]}$ *μπορεί να γραφεί πατώντας τα πλήκτρα* ESC sum ESC CTRL[+] n=0 CTRL[%] m CTRL[␣] 1CTRL/f[n]CTRL[␣] (*τα* [*και*] *δεν τα κτυπάμε*! *το* CTRL[+]*για παράδειγμα σημαίνει* ότι *πατάμε μαζί το* CTRL *και το* + *ενώ το* CTRL[␣] *είναι το* CTRL *μαζί με το* SPACE *πλήκτρο*) ..

Σχόλιο για το Limit: T*o* Limit *αδυνατεί να βρεί* το όριο μιας *σειράς* ό*ταν υπάρχει κάποια σταθερά*:

```
Limit[Sum[a ω^(i-1), {i, 1, n}], n → ∞]
```

$$\text{Limit}\left[\frac{a\,(-1+\omega^{n})}{-1+\omega},\ n \to \infty\right]$$

Για να βρούμε το άθροισμα της σειράς με Limit *θα πρέπει αναγκαστικά να βάλουμε κάποιες τιμές στα* a *και* ω *π.χ*

```
a = 3; w = 0.5; Limit[ Sum[a w^(i-1), {i, 1, n}], n → ∞]

6
```

Άλλος τρόπος: Να αντικαταστήσουμε μέσα στην Sum το n με το ∞(χωρίς να χρησιμοποιήσουμε την Limit).

Mε Sum[f,{i,n,m},{j,k,l}] *μπορούμε να πάρουμε αθροίσματα της* f [i,j] *όταν έχουμε πάνω απο μια μεταβλητή*(*εδώ έχουμε την* i *και την* j) *π.χ.*

```
Sum[x^i y^j, {i, 1, 4}, {j, 1, 2}]

x y + x^2 y + x^3 y + x^4 y + x y^2 + x^2 y^2 + x^3 y^2 + x^4 y^2
```

Εκτός από αθροίσματα μπορούμε να υπολογίζουμε και **γινόμενα**. Η *βασική εντολή που υπολογίζουμε γινόμενα είναι η*

Product[f,{i,imin,imax}]

Μπορούμε να βρούμε και διπλά γινόμενα ή τα όρια για imax->∞ *όπως γίνεται και με το* Sum. *Παραδείγματα*:

```
Product[(i + 1)^2/(i (i + k)), {i, 1, n}]
Product[(i + 1)^2/(i (i + k)), {i, 1, ∞}]

(1 + n) Gamma[1 + k] Gamma[2 + n] / Gamma[1 + k + n]

Gamma[1 + k]
```

Άσκηση: Η *συνάρτηση* φ(x) *του* Euler *δίνει το πλήθος των ακεραίων* n *μεταξύ του* 1 *και του* x *οι οποίοι δεν έχουν κανένα κοινό διαιρέτη με τον* x. *Στο Mathematica η συνάρτηση αυτή δίνεται με την* EulerPhi[x] *π.χ*

```
EulerPhi[60]

16
```

Στη Θεωρία Αριθμών δίνεται με τον τύπο $\varphi(x) = x \prod_{p|x}\left(1 - \frac{1}{p}\right)$*ή με άλλα λόγια είναι το* x *επί το γινόμενο των όρων* $1 - \frac{1}{p}$*όπου το* p *είναι ένας πρωτος διαιρέτης του* x *μικρότερος φυσικά απο το* x. *Να ορίσεται μια συνάρτηση* phiEuler[x_Integer] *που θα είναι ακριβώς το παραπάνω γινόμενο* Ελέγξτε *τα αποτελέσματα σας με την βοήθεια της* EulerPhi.

6.2 Όρια συναρτήσεων μιας μεταβλητής

Έστω f(x) *μια συνάρτηση και θέλουμε να υπολογίσουμε το όριο της καθώς το* x *τείνει στο* x_0. Αυτό *γράφεται στο Mathematica* Limit[$f[x]$, x –> x_0]. Π.χ

```
Limit[(9 x^2 - 1)/(9 x^2 + 51 x - 18), x → 1/3]

2/19
```

Όταν *το όριο δεν υπάρχει* ή *αδυνατεί να το βρεί*, *τότε βγάζει το μήνυμα* Interval[{a,b}] ή *κάτι άλλο.π.χ*

```
Limit[Sin[1/x], x → 0]

Interval[{-1, 1}]
```

Το Interval[{-1,1}] *μας βεβαιώνει* ότι *δεν υπάρχει το* όριο *και* ότι *ίσως υπάρχουν* όρια *υποακολουθιών μεταξύ του* -1 *και* 1. Γενικά όταν *δουλεύομαι με τα* όρια *πρέπει να* είμαστε *προσεκτικοί*. Υπάρχει *δυστυχώς και η περίπτωση που το* Mathematica *δίνει* λάθος όριο. Για *παράδειγμα ας πάρουμε την* $f[x] = \frac{|x-2|}{x-2}$ *που φυσικά παίρνει τιμές* +1 *απο δεξιά και* -1 από *αριστερά*. Οπότε *το* όριο Limit[f[x],x->2] *δεν θα έπρεπε να υπάρχει*. Το *Mathematica* όμως *δίνει λάθος απάντηση*:

```
f[x] := Abs[x - 2]/(x - 2)
Limit[f[x], x → 2]

1
```

Για *τους παραπάνω λόγους θα πρέπει να παίρνουμε μερικές φορές και τα* **πλευρικά όρια** *για να διαπιστώνουμε* με βεβαιότητα *αν παρουσιάζεται κάποια ασυνέχεια* ή *όχι*.Το *όριο απο τα αριστερά το παίρνουμε με* Direction->1 *ενώ απο τα δεξιά με* Direction->-1.

```
Limit[f[x], x → 2, Direction → 1]
Limit[f[x], x → 2, Direction → -1]

-1
```

```
1
```

6.3 Διπλά όρια συναρτήσεων δύο μεταβλητών

Το Mathematica *μας παρέχει την δυνατότητα εύρεσης μόνο των διπλών ορίων* Limit[Limit[f[x,y],y->y0],x->x0] *και* Limit[Limit[f[x,y],x->x0],y->y0]. *Δεν μας παρέχεται η δυνατότητα να βρούμε το* όριο Limit[f[x,y],(x,y)->(x0,y0)](τουλάχιστον για τις εκδόσεις του *Mathematica* που υπάρχουν σήμερα). Τα διπλά *όρια είναι χρήσιμα για να βγάλουμε κάποια συμπεράσματα για την συμπεριφορά μιας συνάρτησης κοντά στο σημείο* (x0,y0). *Για παράδειγμα αν τα όρια* Limit[Limit[f[x,y],y->y0],x->x0] *και* Limit[Limit[f[x,y],x->x0],y->y0] εάν *υπάρχουν αλλά δεν είναι ίσα τότε **δεν** μπορεί να υπάρχει τό διπλό όριο*. ***Αντίστροφα***, γνωρίζουμε απο τα μαθηματικά ότι *αν υπάρχει το όριο τότε τα διπλά είναι ίσα*.
Πριν προχωρήσουμε να εξηγήσουμε ότι για παράδειγμα η εύρεση του διπλού ορίου Limit[Limit[f[x,y],y->y0],x->x0] είναι ισοδύναμη με την εύρεση δυο ορίων. Το πρώτο είναι το εσωτερικό όριο g[x]=-Limit[f[x,y],y->y0](το οποίο είναι συνάρτηση του x) και δεύτερο του εξωτερικού ορίου Limit[g[x],x->x0].
Ας πάρουμε για παράδειγμα την f[x,y]=$\frac{x-y+x^{\wedge}2+y^{\wedge}2}{x+y}$ *και ας βρούμε τα διπλά όρια στο* (x0,y0)= (0,0):

```
Remove[f, x, y]
f[x_, y_] := (x^2 + y)/(x^2 + y^2)
Limit[Limit[f[x, y], x → 0], y → 0]
Limit[Limit[f[x, y], y → 0], x → 0]

∞
```

```
1
```

Θα εξηγήσουμε ότι το πώτο όριο δεν είναι σωστό. Γι αυτό το σκοπό θα χρησιμοποιήσουμε τα *πλευρικά όρια και* την Plot. Οι *γραφικές παραστάσεις μπορούν να μας βοηθήσουν να κατανοήσουμε καλύτερα την συμπεριφορά της* f *γύρω απο ένα σημείο* (x0,y0). Εδώ *δίνονται κάποια παραδείγματα*.

Σχόλιο: Αν *δυσκολεύεστε* κάπου *και ζητάτε βοήθεια, μπορείται να μεταφέρεται τον κέρσορα πάνω σε μια συνάρτηση* ή *σε κάποια επιλογή της(π.χ πάνω στην* Plot) *και μετά πατήστε* F1 *για να πάρετε βοήθεια και παραδείγματα σχετικά με την συνάρτηση που θέλετε βοήθεια*.

```
Remove[f]
f[x_, y_] := If[x^2 + y^2 ≠ 0, (x^2 + y)/(x^2 + y^2), 0]
(* αν ο παρανομαστής είναι 0 τότε θέτουμε τιμή =0*)
x = 0;
ymin = -.2; ymax = .2;
Plot[f[0, y], {y, ymin, ymax},
 PlotRange → {-100, 100}, PlotPoints → 50,
 AxesOrigin → {0, 0}, AspectRatio → .8, MaxBend → 5]
```



- Graphics -

Σχόλια: 1) Βάλαμε *σκόπιμα περισσότερα χαρακτηριστικά* (options) *στην* Plot *απο όσα πράγματι χρειάζονται ώστε να μπορεί κάποιος να μάθει πληροφορίες για αυτά μέσω του πλήκτρου* F1. *Βασικά χρειάζεται η πληροφορία* PlotRange→{-100,100} *γιατί χωρίς αυτή ίσως δεν σας βγεί ενα κατανοητό γράφημα*. *Με την* PlotRange *κόβουμε κατά βούληση τον άξονα των* x ή *των* y. (*εδώ ο* 0y *σχεδιάζεται για τιμές* -100 *έως* 100 *και με* AspectRatio→.8 *ζητάμε το μήκος του κάθετου* πλαισίου που περιβάλλει το σχήμα *να είναι το* 8/10 *του μήκους του οριζόντιου*) Μικρές τιμές της MaxBend (π.χ=5) έχουν σκοπό να κάνουν την καμπύλη πιο "smooth". Χρησιμοποιήσαμε το if στον ορισμό της συνάρτησης διότι η Plot μπορεί να έχει πρόβλημα όταν κάπου εμφανιστεί κλάσμα με παρανομαστή μηδενικό! Αυτό ωφείλεται στο ότι η Plot *παίρνει κάποια σημεία στον άξονα* Ox *βρίσκει τα τιμές της* f *και ενωνει με ευθύγραμμα τμήματα τα σημεία που προκύπτουν*!

2)Επειδή η συνάρτηση $\frac{x^2+y}{x^2+y^2}$ είναι συνεχής ως πρός την x το όριο της καθώς x→0 μπορούμε να το βρούμε απλά θέτοντας x=0 και έτσι παίρνουμε όριο $\frac{1}{y}$. Όμως αυτή η συνάρτηση έχει **διαφορετικά πλευρικά όρια**(-∞, +∞) καθώς το y→0 όπως βλέπουμε και απο το σχήμα. *Ας το δούμε την απάντηση με το* Limit:

```
Limit[Limit[f[x, y], x → 0], y → 0, Direction → 1]
Limit[Limit[f[x, y], x → 0], y → 0, Direction → -1]
```

$$\text{Limit}\left[\text{Limit}\left[\text{If}\left[x^2+y^2\neq 0,\ \frac{x^2+y}{x^2+y^2},\ 0\right],\ y\to 0\right],\ x\to 0,\ \text{Direction}\to 1\right]$$

$$\text{Limit}\left[\text{Limit}\left[\text{If}\left[x^2+y^2\neq 0,\ \frac{x^2+y}{x^2+y^2},\ 0\right],\ x\to 0\right],\ y\to 0,\ \text{Direction}\to -1\right]$$

Δεν τα κατάφερε! Εδώ *δεν φταίει το* Limit *αλλά ο ορισμός με το* If *που δώσαμε για την* f. Το If φαίνεται ότι μπέρδεψε την Limit. Σε τέτοιες περιπτώσεις μπορούμε να χρησιμοποίσουμε την NLimit αφού βέβαια καλέσουμε κάποιο ειδικό πακέτο. Συμβουλευθείτε για λεπτομέρειες το Help. Το Limit φυσικά(εξ ορισμού)*δεν* έχει *πρόβλημα με* κάποια σημεία που μηδενίζουν τον παρανομαστή, *η* Plot όμως *μπορεί να έχει* πρόβλημα λόγω *του τρόπου που σχεδιάζει την γραφική παράσταση*. Ας αντικαταστήσουμε *λοιπόν την* f *με το κλάσμα* $\frac{x\hat{}2+y}{x\hat{}2+y\hat{}2}$*μέσα στο* Limit:

$$\text{Limit}\left[\text{Limit}\left[\frac{x\hat{}2+y}{x\hat{}2+y\hat{}2},\ x\to 0\right],\ y\to 0,\ \text{Direction}\to 1\right]$$
$$\text{Limit}\left[\text{Limit}\left[\frac{x\hat{}2+y}{x\hat{}2+y\hat{}2},\ x\to 0\right],\ y\to 0,\ \text{Direction}\to -1\right]$$

$-\infty$

∞

Άσκηση. Να μελετήσετε την οριακή συμπεριφορά της x Sin[1/y] για (x0,y0)=(0,0).
Λύση. Επειδή -x<= x Sin[1/y]<=x έπεται ότι το όριο υπάρχει και είναι ίσο με 0. Ας δούμε και τα πλευρικά όρια αν πράγματι είναι ίσα.

```
Limit[Limit[x Sin[1 / y], x → 0], y → 0]
Limit[Limit[x Sin[1 / y], y → 0], x → 0]
```

```
0
```

```
Interval[{0, 0}]
```

Βλέπουμε λοιπόν ότι στοMathematica *υπάρχει και η περίεργη περίπτωση να υπάρχει το* όριο *και να "μην υπάρχουν" τα διπλά* όρια(!!!)*Ένα τέτοιο κακό παράδειγμα είναι η συνάρτηση* f[x,y]=x*Sin(1/y). Το *ένα απο τα διπλά* όρια *είναι το* Interval[{0,0}](*δηλ.ουσιαστικά* όριο *το* 0!):

Ας κάνουμε και εδώ την γραφική παράσταση για αρκετές τιμές του y *κοντά στο* 0 *για να καταλάβουμε την* περίεργη *απάντηση* Interval[{0,0}].

```
Remove[x, y, z, g];
g[x_, y_] := If[y ≠ 0, x Sin[1 / y], 0]
x1 = -1; x2 = 1;
pinakas = Table[g[x, y], {y, -.02, 0, 1 / 400}]
(*πινακας με  9 τιμες της g[x,y] με y
  πολύ κοντά στο 0 και από τα αριστερά του 0*)
Plot[Evaluate[pinakas], {x, x1, x2}, PlotRange → All,
 AspectRatio → 1, PlotPoints → 40]

{0.262375 x, -0.559837 x, 0.639018 x, 0.993889 x,
 0.506366 x, -0.983055 x, 0.873297 x, 0.850919 x, -0.831475 x}
```



- Graphics -

απο αυτά βλέπουμε ότι *οι τιμές του εσωτερικού ορίου* Limit[x Sin[1/y],y→0] *δεν τείνουν σε ένα όριο* f[x] *δηλαδή σε κάποια* συγκεκριμμένη *συνάρτηση* g *του* x *(βλέπε και* pinakas)*με αποτέλεσμα να είναι "αδύνατο" να βρεθεί και το εξωτερικό* όριοLimit[g[x],x→0]= Limit[Limit[x Sin[1/y],y→0],x→0]. *Έτσι άλλοτε παίρνουμε* όριο 0 *απο δεξιά* (0^+) *και άλλοτε* 0 *απο αριστερά*(0^-).

Σχόλιο για την Evaluate. Η Evaluate μέσα *στην* Plot *αναγκάζει να υπολογιστούν* ***πρώτα*** όλες *οι συναρτήσεις του* pinaka *πριν εφαρμοστεί η* Plot για να σχεδιάσει κάθε μια απο αυτές. *Διαφορετικά η* Plot *δεν θα μπορούσε να κάνει την γραφική παράσταση διότι θα νόμιζε* ότι *ο* pinaka *είναι* l;ista κάποιων συμβόλων *και* όχι *κάποιες συναρτήσεις*!)

Άσκηση: Αφού *βρείτε απο το* Help *του Mathematica τι κάνει* το παρακάτω πακέτο <<Graphics`Animation` και η εντολή Animate, *να ερμηνεύσεται με τις γραφικές παραστάσεις που παράγονται, την συμπεριφορά των διπλών ορίων της* g[x_,y_]:=If[y≠0,x Sin[1/y],0]. (*προσοχή αν πατήσετε διπλό κλικ σε κάποια απο τις γραφικές παρστάσεις που θα προκύψουν θα δείτε* Movie!) Το πρώτο Animate υπολογίζει "γραφικά" το όριο Limit[Limit[x Sin[1/*y*], $y \to 0^-$],x→0] ενώ το δεύτερο το όριο Limit[Limit[x Sin[1/*y*], $x \to 0^-$],y→0]. Βάλαμε -0.0005 αντί 0 γιατί η Plot θα βγάλει μήνυμα λάθους όταν θα αρχίσει να σχεδιάζει μια συνάρτηση με Πεδίο όρισμού ένα μόνο σημείο (το 0).

```
<< Graphics`Animation`
g[x_, y_] := If[y ≠ 0, x Sin[1 / y], 0]
Animate[Plot[g[x, y],
   {x, y, -y}, PlotRange → {{-1, 1}, {-1, 1}}], {y, -1, -0.0005}];
```

```
Animate[Plot[g[x, y],
   {y, x, -x}, PlotRange → {{-0.5, 0.5}, {-1, 1}}],
  {x, -1, -.0005}];
```

6.4 Ακρότατα συναρτήσεων

Στη μελέτη των συναρτήσεων συμπεριλαμβάνεται και η εύρεση των τοπικών ακροτάτων. *Για να βρούμε ένα τοπικό ελάχιστο της* f[x] *γύρω απο το* x_0 *γράφουμε* FindMinimum[$f[x]$, {x, x_0}]. *Για το τοπικό μέγιστο αρκεί να ζητήσουμε το ελάχιστο της* -f[x] *διότι* ως γνωστό, maxf= -min[-f]. *Αν βέβαια* έχουμε *συνάτηση περισότερων μεταβλητών τότε γίνονται οι κατάλληλες αλλαγές π.χ για να βρούμε το τοπικό ελάχιστο γύρω από το σημείο* (x_0, y_0) *γράφουμε* FindMinimum[$f[x, y]$, {x, x_0}, {y, y_0}]. *Ας δούμε τις συναρτήσεις αυτέςσε σχέση με την* g *που ήδη μελετήσαμε* στην προηγούμενη ενότητα. Ας βρούμε πρώτα τα ακρότατα ως προς την y.

```
FindMinimum[g[.05, y], {y, 0.5}]
-FindMinimum[-g[.05, y], {y, 0.5}]

{-0.05, {y → 0.212207}}
```

```
{0.05, {-(y → 0.63662)}}
```

Δηλαδή για x0=0.05 *έχουμε μέγιστο* 0.05 *για* y=0.63662 *και ελάχιστο*-0.05 *για* y=0.212207.*Ας κάνουμε και την γραφική παράσταση*:

```
ymin = 0; ymax = 1;
Plot[g[.05, y], {y, ymin, ymax},
 PlotRange → All, AspectRatio → 1, PlotPoints → 40]
```




```
- Graphics -
```

Αν προσπαθήσουμε να βρούμε μέγιστο και ελάχιστο της g *για τιμές του* y *σε ένα διάστημα* [ymin,ymax] *θα πρέπει να βάλουμε τα όρια που θα κινείται το* y *ως εξής*

```
y0 = 0.005; ymin = -1; ymax = .01;
FindMinimum[g[0.05, y], {y, y0, ymin, ymax}]
-FindMinimum[-g[0.05, y], {y, y0, ymin, ymax}]
```

```
{-0.05, {y → 0.00501275}}
```

```
{0.05, {-(y → -0.212207)}}
```

Απο αυτά βλέπουμε ότι η εύρεση του τοπικού μεγίστου και ελαχίστου εξαρτάται απο το σημείο που δίνομε στο FindMinimum. *Έτσι για* y_0 =0.5 *έδωσε άλλα ακρότατα και για* y_0 =0.005 *άλλα.*

```
Plot[g[0.05, y], {y, -0.3, .01}, PlotRange → All, AspectRatio → 1,
 PlotPoints → 40, AxesStyle → {{}, {RGBColor[1, 0, 0]}}]
```




```
- Graphics -
```

Στην παραπάνω γραφική παράσταση ζωγραφίσαμε τον κάθετο άξονα με κόκκινο για να τον εντοπίσετε. Το y=-0.212207 μπορούμε να το εντοπίσουμε στα αριστερά αλλά το 0.00501275 είναι αδύνατο να το εντοπίσουμε διότι πραγματικά η συνάρτηση g κοντά στο 0 ανεβοκατεβαίνει με τρελούς ρυθμούς...

Εύρεση ακροτάτων ως προς την x:

Αν τώρα σταθεροποιήσουμε την τιμή του y ***π.χ*** y=.05 και θεωρήσουμε σαν μεταβλητή την x ***βλέπουμε*** ***ότι είναι αδύνατον να πάρουμε κάποια ακρότατα***:

```
x0 = 0.05; x1 = -1; x2 = 1;
FindMinimum[g[x, .05], {x, x0, x1, x2}]
```

```
FindMinimum::regex : Reached the point -1. which is outside the region {{-1., 1.}}.
```

```
FindMinimum[g[x, 0.05], {x, x0, x1, x2}]
```

Δηλαδή δεν μπόρεσε να βρεί ένα τοπικό ελάχιστο της g ***στα δοθέντα όρια του*** x και για y=0.05 . ***Ας κάνουμε πάλι την γραφική παράσταση για να δούμε τι ακριβώς συμβαίνει***:

```
Remove[x]
g[x, 0.05]
Plot[g[x, 0.05], {x, xmin, xmax},
 PlotRange → All, AspectRatio → 1, PlotPoints → 40]
```

```
0.912945 x
```




```
- Graphics -
```

πράγματι λοιπόν δεν έχει κάποιο ακρότατο(μέγιστο ή ελάχιστο) ***αλλά οι τιμές*** g[x,0.05]***μεγαλώνουν*** ή ***μικραίνουν συνέχεια***!
Τέλος, αν προσπαθήσουμε να βρούμε ***τα ακρότατα και ως προς τις δύο μεταβλητές θα διαπιστώσουμε τα ίδια***:

```
x0 = 0.05; x1 = -1; x2 = 1; y0 = 0.05; ymin = -1; ymax = .01;
FindMinimum[g[x, y], {x, x0, x1, x2}, {y, y0, ymin, ymax}]
```

```
FindMinimum::regex :
 Reached the point {0.05, 0.05} which is outside the region {{-1., 1.}, {-1., 0.01}}.
```

```
FindMinimum[g[x, y], {x, x0, x1, x2}, {y, y0, ymin, ymax}]
```

<u>Άσκηση</u>: Χρησιμοποιείστε την FindMinimum για να βρείτε αν υπάρχει το όριο της g[x,y] γύρω απο το σημείο $(x_0, y_0) = (0.05, 0.5)$.

Άσκηση Στις νεότερες εκδόσεις του προγράμματος υπάρχουν και οι συναρτήσεις υπάρχουν οι και οι FindMaximum , Maximize, NMaximize, Minimize, NMinimize. Δείτε τι ακριβώς κάνει η καθεμία και πειραματιστείτε...

6.5 *Σειρές δυνάμεων, Σειρές* Taylor *και* Mac-Laurin

Έστω f(x) *συνάρτηση, που έχει συνεχείς παραγώγους ως προς* x *μέχρι και* n *τάξεως στο σημείο* a *και υπάρχει η παράγωγος* n+1 *τάξης της* f *στο* a *τότε υπάρχει το ανάπτυγμα της* f *σε σειρά γύρω απο το* a *σε δυνάμεις* $(x-a)^n$ *π.χ. για* n=4

```
Series[f[x], {x, a, 4}]
```

$$f[a] + f'[a]\,(x-a) + \frac{1}{2} f''[a]\,(x-a)^2 + \frac{1}{6} f^{(3)}[a]\,(x-a)^3 + \frac{1}{24} f^{(4)}[a]\,(x-a)^4 + O[x-a]^5$$

Το n+1=5 *και το* $O[x-a]^5$ *παριστάνει το* ***υπόλοιπο***(ή *σφάλμα*) n+1 *τάξης της* f *για το σημείο* a *και έχει την ιδιότητα το όριο καθώς* x->*α να είναι* 0 *δηλ*. Limit[$O[x-a]^5$,x->a]=0. Ο *παραπάνω τύπος είναι ο τύπος του* Taylor *για την* f. *Ειδικά όταν* a=0 *ο τύπος αυτός γίνεται*

```
Series[f[x], {x, 0, 4}]
```

$$f[0] + f'[0]\,x + \frac{1}{2} f''[0]\,x^2 + \frac{1}{6} f^{(3)}[0]\,x^3 + \frac{1}{24} f^{(4)}[0]\,x^4 + O[x]^5$$

που δεν είναι άλλος απο τον τύπο του **Mac Laurin** *για την* f. *Αλλά ας δούμε και παραδείγματα*:

```
seira = Series[Log[x], {x, 1, 4}]
```

$$(x-1) - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 - \frac{1}{4}(x-1)^4 + O[x-1]^5$$

Την παραπάνω σειρά μπορούμε να την υψώσουμε στο τετράγωνο να την παραγωγίσουμε ως προς x *να την ολοκληρώσουμε π.χ*

```
seira^2
D[seira, {x, 2}](*η δευτερη παράγωγος ως προς x *)
Integrate[seira, x](*αόριστο ολοκληρωμα ως προς x *)
```

$$(x-1)^2-(x-1)^3+\frac{11}{12}(x-1)^4-\frac{5}{6}(x-1)^5+O[x-1]^6$$

$$-1+2(x-1)-3(x-1)^2+O[x-1]^3$$

$$\frac{1}{2}(x-1)^2-\frac{1}{6}(x-1)^3+\frac{1}{12}(x-1)^4-\frac{1}{20}(x-1)^5+O[x-1]^6$$

Χρήσιμες και στενά συνδεμένες με την Series *είναι και οι* Normal *που αποκόπτει το υπόλοιπο* $O[x-a]^n$ *απο το ανάπτυγμα* Taylor *και η* SeriesCoefficient *που επιστρέφει τον συντελεστή κάποιας δύναμης* $(x-a)^m$ *στο ανάπτυγμα* Taylor (όμως θα πρέπει η συνάρτηση που δουλεύομαι να είναι μιας μεταβλητής) . *Ας δούμε τα παραδείγματα που ακολουθούν* :

```
Normal[seira^2]
SeriesCoefficient[seira^2, 4]
Coefficient[Normal[seira^2], x - 1, 4]
```

$$(-1+x)^2-(-1+x)^3+\frac{11}{12}(-1+x)^4-\frac{5}{6}(-1+x)^5$$

$$\frac{11}{12}$$

$$\frac{11}{12}$$

Απο τα παραπάνω βλέπουμε ότι η Coefficient σε συνδυασμό με την Normal μπορεί να αντικαταστήσει άνετα την SeriesCoefficient.

Τελειώνοντας να πούμε ότι *αν η συνάρτηση* f *είναι δυο μεταβλητών τότε μπορούμε να πάρουμε την* ***διπλή*** Series. *Έτσι η* `Series[f, {x, x0, nx}, {y, y0, ny}]` *βρίσκει πρώτα το ανάπτυγμα ως προς το x, και μετά ως προς το y.*

Π.*χ*

```
Series[Sin[x y], {x, 0, 7}, {y, 0, 7}]
```

$$(y + O[y]^8)\,x + \left(-\frac{y^3}{6} + O[y]^8\right)x^3 + \left(\frac{y^5}{120} + O[y]^8\right)x^5 + \left(-\frac{y^7}{5040} + O[y]^8\right)x^7 + O[x]^8$$

6.6 Η εφαπτόμενη μιας συνάρτησης

Με *την ευκαιρία των σειρών* Taylor *της* f *γύρω απο το σημείο* a *πρέπει να πούμε* ότι *τα πολυωνυμικά αναπτύγματα που προκύπτουν με την συνάρτηση* Normal *προσεγγίζουν πολύ καλά την συνάρτηση* f *γύρω απο το* a. ¨Οσο *μεγαλύτερο το* n *τόσο καλύτερη η προσέγγιση*. *Ουσιαστικά λοιπόν τα αναπτύγματα* Taylor *αποτελούν τα* **"εφαπτόμενα" πολυώνυμα** *της* f *στο* a. Για n=1 έ*χουμε την* **εφαπτόμενη ευθεία** Παράδειγμα: *για να βρούμε την εφάπτομενη στην* Log[x] *για α*=0.5 *θέτουμε* n=1 *στην* Series *και μετά την παίρνουμε την* Normal

```
efaptomenh = Normal[Series[Log[x], {x, .5, 1}]]
```

-0.693147 + 2. (-0.5 + x)

```
Plot[{Log[x], efaptomenh}, {x, 0, 2},
 AxesOrigin → {0, 0}, AspectRatio → 1, PlotStyle →
  {{RGBColor[1, 0, 0], Thickness[.01]}, {Dashing[{0.02, 0.02}]}}]
```




```
- Graphics -
```

Σχόλιο για την Plot. *Στο παραπάνω γράφημα σχεδιάσαμε μαζί την* Log *και την εφαπτόμενη ευθεία*. Το RGBColor[1,0,0] *είναι το κόκκινο χρωμα και το χρησιμοποιήσαμε για την* Log *ενώ το* Dashing *δηλ. τις διακεκομμένες γραμμές για την εφαπτομένη*. Παρατηρείστε *ότι πράγματι η ευθεία είναι εφαπτόμενη στο σημείο* a=0.5.

Με *χρήση της* Normal[Series[Log[x],{x,.5,n}]] *για* n *μεγαλύτερα του* 1 *μπορούμε να πάρουμε καλύτερες προσεγιστικές καμπύλες στην* Log. Μπορείται *να κάνεται την γραφική παράσταση για* n>1 *για να δείτε την διαφορά*. Αν *τώρα θέλουμε μια καλύτερη προσέγγιση με πολυώνυμο π.χ* 4 *βαθμού* (*γύρω απο το σημείο* a *φυσικά*)*θα πρέπει να βρούμε την* Series 4 *βαθμού δηλαδή*:

```
efaptomenh4 = Normal[Series[Log[x], {x, .5, 4}]]
```

$$-0.693147 + 2.\ (-0.5 + x) - 2.\ (-0.5 + x)^2 + 2.66667\ (-0.5 + x)^3 - 4.\ (-0.5 + x)^4$$

```
Plot[{Log[x], efaptomenh4}, {x, 0, 2},
 AxesOrigin → {0, 0}, AspectRatio → 1, PlotStyle →
  {{RGBColor[1, 0, 0], Thickness[.01]}, {Dashing[{0.02, 0.02}]}}]
```



- Graphics -

Άσκηση: *Δίνεται η συνάρτηση* f[x_]:=$\sqrt{2+x}$. Βρείτε *τα αναπτύγματα* Mac-Laurin *της* f[x] *βαθμού* n=1,2,...20 *με την βοήθεια της* Series *και ομαδοποιήστε τα αποτελέσματα σε μια λίστα. Στην συνέχεια εφαρμόσετε την* Normal *στη λίστα και στην λίστα που προκύπτει θέστε* x=0. *Θα προκύψουν* 20 *αριθμοί.* Εξηγείστε *γιατί αυτοί οι όροι αποτελούν μια συγλίνουσα ακολουθία με όριο το* $\sqrt{2}$. Πόσο *περίπου είναι το* $\sqrt{2}$ *απο τις προσεγγίσεις σας;*